Example 7c – Plate perforation (Courtesy of CEA)

- A tough cylindrical projectile impacts a ductile circular plate at the initial velocity of 100 m/s and completely perforates it



57

TITLE:
Plate perforation problem.
PROBLEM:
This problem has been studied at CEA. An elastoplastic punch impacts a circular metallic plate at an initial velocity of 100 m/s and perforates it. The punch material is quite resistant and is only slight deformed, while the plate material is rather ductile and undergoes very large plastic strain, culminating into complete failure along the border of the zone in contact with the punch.
MESH:
Although the physical problem is 2D axisymmetric (both in the geometry and in the loading), a full 3D numerical model is assumed for the calculation, made of 55848 hexahedra (CUBE) elements and 62275 nodes. As many as twelve layers of elements are taken across the plate thickness, and the plate mesh is very refined in the region of contact between the plate and the punch, in order to represent plasticity and material failure in a rather accurate manner. Only ¼ of the geometry is modelled, by assuming two symmetry planes in *x-z* and *y-z*.
MATERIALS:
The VMIS ISOT (Von Mises law with isotropic hardening) material is assigned to the punch, while the plate has the LEM1material (Lemaitre elasto-plastic material with damage). In the latter material, there is coupling between damage and plasticity, represented vy the Von Mises criterion. The damage evolution rate is a function of the triaxiality ratio of stresses and of the equivalent plastic strain rate. A failure criterion is implicitly contained within the model: rupoture occures when the damage exceeds a critical value.
BOUNDARY CONDITIONS:
All nodes on the external circumference of the plate are blocked in all three space dimensions, so as to simulate plate clamping around the external perimeter. Two symmetry planes, along *x-z* and *y-z*, respectively, are specified as already mentioned above. Finally, the GLIS model is used to descrive the contact between the punch extremity and the central part of the plate.
LOADING:
No external loads are assumed.
INITIAL CONDITIONS
The punch has an initial velocity of 100 m/s in the vertical direction towards the plate.
CALCULATION:
The calculation is performed up to 1 ms. At the final time, the punch has completely perforated the plate.

POST-TREATMENT

A detailed animation of the computed results from this calculation is available on the EUROPLEXUS Consortium Web site.

<u>Numerical Solutions</u>

PERF01

3D solution (1/4 model). The mesh generation file is:

```
*%siz 300
*
opti echo 1;
OPTI DIME 3 ELEM CUB8 ;
OPTI SORT  'perf01.msh' ;
opti trac psc ftra 'perf01_mesh.ps';
*
*++++++  maillage de la plaque
*
REP1 = @REPERE (0. 0. 0.) (prog 0.3 0.3 0.15) turq ;
e_plaq   = 0.006 ;
r_plaq   = 0.225 ;
n_plaq   = 50 ;
n_couch1 = 12 ;
h_poin   = (e_plaq + 0.050) ;
r_poin   = 0.024 ;
n_poin   = 6 ;
n_couch2 = 6 ;
o_0  =                     0.                      0.  0. ;
p1_0e =                r_plaq                      0.  0. ;
p2_0e =     (r_plaq*(cos 45))      (r_plaq*(sin 45))  0. ;
p3_0e =                     0.                 r_plaq  0. ;
p1_0i =             (r_plaq/6)                     0.  0. ;
p2_0i = ((1.2*r_plaq/6)*(cos 45))  ((1.2*r_plaq/6)*(sin 45))  0. ;
p3_0i =                     0.             (r_plaq/6)  0. ;
c12_0e = CERC n_plaq p1_0e o_0 p2_0e ;
c23_0e = CERC n_plaq p2_0e o_0 p3_0e ;
c13_0e = c12_0e ET c23_0e ;
d01_0i = o_0   DROI n_plaq p1_0i ;
d12_0i = p1_0i DROI n_plaq p2_0i ;
d23_0i = p2_0i DROI n_plaq p3_0i ;
d30_0i = p3_0i DROI n_plaq o_0 ;
*
* maillage du centre
*
s_centr = DALLER d01_0i d12_0i d23_0i d30_0i PLAN ;
v_plaq  = 0. 0. e_plaq ;
v_centr = (s_centr VOLU n_couch1 TRAN v_plaq) coul roug ;
* maillage de la partie extérieure
c13_0i  = d12_0i ET d23_0i ;
s_exter = c13_0i REGLE 'DINI' 0.002 'DFIN' 0.02 c13_0e ;
v_exter = (s_exter VOLU n_couch1 TRAN v_plaq) coul roug ;
plaq = v_centr et v_exter  ;
plaq = plaq coul roug ;
elim plaq 0.0001 ;
trac (plaq et rep1) face ;
*
*++++++  maillage du poinçon plat
*
o_3  =                    0.                     0.  (e_plaq + 0.001) ;

p1_3e =               r_poin                     0.  (e_plaq + 0.001) ;
p2_3e =     (r_poin*(cos 45))      (r_poin*(sin 45))  (e_plaq + 0.001) ;
p3_3e =                    0.                 r_poin  (e_plaq + 0.001) ;

p1_3i =            (r_poin/2)                    0.  (e_plaq + 0.001) ;
p2_3i = ((r_poin/2)*(cos 45))  ((r_poin/2)*(sin 45))  (e_plaq + 0.001) ;
p3_3i =                    0.            (r_poin/2)  (e_plaq + 0.001) ;

c12_3e = CERC (n_poin) p1_3e o_3 p2_3e ;
c23_3e = CERC (n_poin) p2_3e o_3 p3_3e ;
c13_3e = c12_3e ET c23_3e ;

d01_3i = o_3   DROI (n_poin) p1_3i ;
d12_3i = p1_3i DROI (n_poin) p2_3i ;
d23_3i = p2_3i DROI (n_poin) p3_3i ;
d30_3i = p3_3i DROI (n_poin) o_3 ;

s0123_3i = DALLER d01_3i d12_3i d23_3i d30_3i PLAN ;

c13_3i = d12_3i ET d23_3i ;
s123_3 = c13_3i REGLE (n_poin) c13_3e ;

surf_3 = s0123_3i ET s123_3 ;

v_poin = 0. 0. h_poin ;
poinc  = surf_3 VOLU n_couch2 TRAN v_poin ;
poinc  = poinc COUL VERT ;
elim poinc 0.0001 ;

lig1    = (r_plaq 0. 0.) droi n_couch1 (r_plaq 0. e_plaq) ;
b_plaq  = rota lig1 n_plaq 90. (0. 0. 0.) (0. 0. 1.) ;

allmesh =  plaq et poinc et b_plaq ;
elim allmesh 0.0001 ;

TRAC (plaq et poinc et rep1) CACH face ;

pts_xoz = allmesh POIN PLAN (0. 0. 0.) (1. 0. 0.) (0. 0. 1.) 0.0005 ;
pts_zoy = allmesh POIN PLAN (0. 0. 0.) (0. 1. 0.) (0. 0. 1.) 0.0005 ;

sort allmesh ;

mess 'nb de noeuds =' (nbno allmesh) ;
mess 'nb d elements =' (nbel allmesh) ;
mess 'volume du poinçon =' (mesure poinc volu) 'm3';

fin;
```

The input file is:

```
plaque poinçon plat - Essai P7, M = 496 Kg, Vini = 100 m/s
*
ECHO
 conv win
*
TRID NONL LAGC FAIL
*
GIBI allmesh
*
DIME
  PT3L 62275 CUBE 55848 ZONE 1
  BLOQ 1989
  LIAI 14661
  GLIS 1 200000
  ECRO 778632
  SYME 2
TERM
*
GEOM CUBE v_centr v_exter poinc TERM
*
MATE VMIS ISOT RO 4926245.4 YOUN 2.E13  NU 0.3  ELAS 300.E6
          TRAC 2
                  0.300E+09  1.5E-05
                  0.500E+09  45E-02
          LECT poinc TERM
     LEM1 RO 7850 YOUN 174410.E6 NU 0.3
          ELAS 166.5E6 EPSD 0. S0 16390295.E+12 DC 0.01 TRAC 96
             .1665E+09    .954704E-03
             .1764E+09    .1017E-02
             .2228E+09    .1374E-02
             .2597E+09    .1830E-02
             .2824E+09    .2404E-02
             .2952E+09    .3050E-02
             .3037E+09    .3763E-02
             .3090E+09    .4499E-02
             .3146E+09    .5263E-02
             .3183E+09    .6020E-02
             .3214E+09    .6766E-02
             .3254E+09    .7551E-02
             .3284E+09    .8324E-02
             .3309E+09    .9096E-02
             .3339E+09    .9870E-02
             .3365E+09    .1065E-01
             .3385E+09    .1142E-01
             .3406E+09    .1219E-01
             .3428E+09    .1297E-01
             .3451E+09    .1375E-01
             .3476E+09    .1453E-01
             .3486E+09    .1530E-01
             .3505E+09    .1578E-01
             .3545E+09    .1731E-01
             .3723E+09    .2502E-01
             .3873E+09    .3263E-01
             .4015E+09    .4008E-01
             .4161E+09    .4740E-01
             .4301E+09    .5461E-01
             .4441E+09    .6172E-01
             .4576E+09    .6875E-01
             .4700E+09    .7566E-01
             .4821E+09    .8246E-01
             .4954E+09    .8921E-01
             .5066E+09    .9584E-01
             .5189E+09    .1024E+00
             .5288E+09    .1089E+00
             .5407E+09    .1153E+00
             .5513E+09    .1216E+00
             .5620E+09    .1279E+00
             .5720E+09    .1341E+00
             .5816E+09    .1402E+00
             .5918E+09    .1462E+00
             .6004E+09    .1522E+00
             .6104E+09    .1582E+00
             .6246E+09    .1690E+00
             .6318E+09    .1752E+00
             .6411E+09    .1814E+00
             .6494E+09    .1875E+00
             .6581E+09    .1937E+00
             .6682E+09    .1998E+00
             .6767E+09    .2058E+00
             .6848E+09    .2118E+00
             .6932E+09    .2178E+00
             .7021E+09    .2237E+00
             .7096E+09    .2296E+00
             .7159E+09    .2355E+00
             .7256E+09    .2413E+00
             .7330E+09    .2471E+00
             .7421E+09    .2528E+00
             .7491E+09    .2586E+00
             .7564E+09    .2643E+00
             .7634E+09    .2700E+00
             .7721E+09    .2756E+00
             .7800E+09    .2812E+00
             .7865E+09    .2867E+00
             .7922E+09    .2923E+00
             .8006E+09    .2978E+00
             .8098E+09    .3033E+00
             .8159E+09    .3087E+00
             .8228E+09    .3142E+00
             .8287E+09    .3195E+00
             .8700E+09    .3500E+00
             .9338E+09    .4000E+00
             .1051E+10    .5000E+00
             .1188E+10    .6300E+00
             .1312E+10    .7600E+00
             .1427E+10    .8900E+00
             .1534E+10    .1020E+01
             .1634E+10    .1150E+01
             .1730E+10    .1280E+01
             .1821E+10    .1410E+01
             .1908E+10    .1540E+01
             .1992E+10    .1670E+01
             .2072E+10    .1800E+01
             .2150E+10    .1930E+01
             .2226E+10    .2060E+01
             .2299E+10    .2190E+01
             .2371E+10    .2320E+01
             .2440E+10    .2450E+01
             .2508E+10    .2580E+01
             .2574E+10    .2710E+01
             .2639E+10    .2840E+01
             .2702E+10    .2970E+01
             .2764E+10    .3100E+01
             .2825E+10    .3230E+01
          LECT v_centr v_exter TERM
*
LIAI BLOQ 123 LECT b_plaq TERM
     GLIS 1 MAIT LECT v_centr TERM
            ESCL LECT poinc TERM
     CONT SPLA NX  0 NY -1. NZ 0. LECT pts_xoz TERM
          SPLA NX -1 NY  0. NZ 0. LECT pts_zoy TERM
*
INIT VITE 3 -100 LECT poinc TERM
          3    0 LECT plaq  TERM
*
ECRI TFRE 1.E-6 NOPO NOEL
     FICH ALIC TFRE 1.E-5
     FICH K200 TFRE 10.E-5
          POIN TOUS CHAM
OPTI csta 0.5
     log 1
CALC TINI 0. TEND 100.E-5 NMAX 999999
FIN
```

The configurations at 0.5 ms (half-time) and 1.0 ms (final time) are:



The same configurations, more detailed:





The first ruptured elements appear at time 0.2 ms (seen from below) and the final configuration:





The plastic strain in the plate at mid-time and at final time is:





The same results, more detailed, are:





The plastic strains and the mesh, seen from below, are shown at time 0.2 ms when the first elements fail and the same at 0.5 ms:





The final view is: